Fermi National Accelerator Laboratory

FERMILAB-TM-1944

E872

Groundwater Activation Calculations for E872

W.S. Freeman
The E872 Collaboration

Fermi National Accelerator Laboratory
P.O. Box 500, Batavia, Illinois 60510

August 1995

Operated by Universities Research Association Inc. under Contract No. DE-AC02-76CHO3000 with the United States Department of Energy

Disclaimer

Fermilab

Groundwater Activation Calculations for E-872

William S. Freeman
July 28, 1995

The E872 beam dump geometry has been modeled in CASIM and calculations have been done to determine the annual limits for protons on target. Results are presented using both the single resident well model (SRWM) and the newly-approved concentration model (CM). The conclusion is that the target/dump design is adequate for the maximum number of protons on target requested by the experiment , which is $>1x10^{18}$ protons per year at 800 GeV.

Description of Modeled Geometry
The E872 target/beam dump will be located near the upstream end of the PW8 experimental hall. The target/dump was modeled as a 1 meter long rectangular tungsten block with transverse dimensions 10 cm (h) x 5 cm (w). It was recessed in surrounding steel shielding with its upstream end located 1 meter inside the shield. Because the beam elevation at the target location is only 91 cm above the floor it was decided to support the target from above to allow the maximum possible shielding beneath the target. Thus the target was assumed to reside in a cavity with internal dimensions of 26 cm (h) x 20 cm (w) with only a 1 cm clearance between the bottom of the target and the steel shielding (see figure 1). The width of the cavity was determined by the need to use a "half-density" tungsten target for part of the running period with provision to move it into the beam remotely. This half-density target was modeled as five tungsten blocks, each of dimension 10 cm (h) x 5 cm (w) x 10 cm (l), equally spaced beside the full density target (see figure 2). The beam was assumed to be targeted on the center of the upstream face of the full-density target for the results reported here.

A large steel sweeping magnet (SELMA) will be located immediately downstream of the target/dump. The magnet aperture will be completely filled. For simplicity this magnet was modeled as a solid, unmagnetized iron block with dimensions 190 cm (h) x 290 cm (w) x 700 cm (l), centered on the beam line. (see figure 3). The beam elevation above the floor increases from 91 to 182 cm beginning at the upstream end of the SELMA magnet. The magnet rested on a 45 cm thick concrete block placed on the floor of the hall. An extra 30 cm of steel was added to the upper surface of the magnet to reduce residual activation rates to acceptable levels.

A second sweeping magnet (MuSweep 2) will follow downstream of SELMA. It has a "picture-frame" cross section with one vertical side centered on the beam axis and is 500 cm long. It also was modeled as solid, unmagnetized iron with cross-section dimensions shown in figure 4.

For completeness, the downstream passive steel muon absorbers were also included in the modeled geometry with dimensions as shown in figure 5, although the bulk of the stars and the highest star density in the soil surrounding the enclosure are expected to be concentrated in the vicinity of the target and the SELMA magnet and not around the muon absorbers.

The PW8 enclosure was modeled with 45 cm thick concrete walls, floor, and ceiling. The upstream region of the hall had a rectangular cross section with inner dimensions shown in figure 6a and 6b. Note that the beam elevation is 91 cm above the floor in the upstream region. The downstream region has a larger cross section due to a lower floor elevation. The enclosure was assumed to be surrounded by soil out to a radius of 750 cm which ensured a minimum of one meter of earth in all radial directions.

Plan and elevation views of the full modeled geometry are shown in figures 7a and 7b. The CASIM FORTRAN geometry file is reproduced in Appendix 1. The CASIM input data file is shown in Appendix 2.

Results

Single Resident Well Model (SRWM)

The total number of stars produced in the soil surrounding the PW8 enclosure was calculated to be 0.10 ± 0.02. This was obtained from a CASIM run using 100,000 incident particles at 800 GeV. The final concentration of the ith radionuclide C_i^{final} (in pCi /ml-yr) is given by

$$C_i^{final} = \frac{N_p \cdot S_T \cdot K_i \cdot L_i \cdot e^{(-y/v_i \tau_i)}}{\tau_i \cdot 6.47 \times 10^{13}} \qquad \text{(eq. 1)}$$

where

N_p	is the average number of incident protons per year
S_T	is the total number of stars per proton produced in the unprotected soil region
$K_i \cdot L_i$	is the number of leachable atoms of the ith nuclide produced per star (0.0675 atoms/star for H^3, 0.0027 atoms /star for Na^{22})
τ_i	the mean decay life for the ith nuclide (17.7 years for H^3, 3.74 years for Na^{22})
v_i	the vertical velocity of the ith nuclide (2.19 m/yr for H^3, 0.98 m/yr for Na^{22})
y	the vertical distance from the source to the aquifer (11.3 meters)

6.47×10^{13} conversion factor to get proper units and based on the standard 40 gallons per day pumped from a single well for an entire year

The sum of the ratios of concentrations to their allowed regulatory limits must be less than 1 to insure that the annual 4 mrem per year limit for community drinking water supplies is not exceeded (ref: 40 CFR 141). That is,

$$\frac{C(H^3)}{20} + \frac{C(Na^{22})}{0.2} \leq 1 \qquad \text{(eq. 2)}$$

Using equations 1 and 2, the maximum permitted number of protons on target per year was determined to be 2.0×10^{18}. The results are summarized in Table 1.

Concentration Model (CM)

The recently approved concentration model (ref: TM1851, Malensek, et al and Environmental Protection Note 8, Cossairt) also was used to determine the allowed number of protons on target per year.

In the concentration model the initial concentration C_i, for radionuclide, i, is given by

$$C_i = \frac{N_p \cdot S_{\max} \cdot 0.019 \cdot K_i \cdot L_i}{1.17 \times 10^6 \cdot \rho \cdot w_i} \qquad \text{(eq. 3)}$$

where

N_p is the number of incident protons per year

$S_{\max}$ is the maximum star density per incident proton in the unprotected soil

K_i is the radionuclide production probability per star (0.075 atoms/star for H^3, 0.02 atoms /star for Na^{22})

L_i is the leachability factor for the radionuclide (0.9 for H^3 and 0.135 for Na^{22})

ρ is the soil density (2.25 gm/cm^3 for moist soil)

w_i is the weight of water divided by the weight of soil that corresponds to 90% leaching (0.27 for H^3 and 0.52 for Na^{22})

The final concentration in groundwater, C_i^{final}, is related to the initial concentration by

$$C_i^{final} = C_i \cdot R_{till} \cdot R_{mix} \cdot R_{dolomite} \qquad \text{(eq. 4)}$$

where R_{till} is a reduction factor in the model that takes account of the migration of radionuclides downward through the soil (glacial till) to the aquifer and allows for radioactive decay *en route*. The additional reduction factors, R_{mix} and $R_{dolomite}$ are assumed equal to one. The reduction factors are given by

$$R_{till}\left(H^3\right) = 1.0 \cdot e^{[-0.3 \cdot d(meters)]} \qquad \text{(eq. 5)}$$

$$R_{till}\left(Na^{22}\right) = 1.0 \cdot e^{[-0.92 \cdot d(meters)]} \qquad \text{(eq. 6)}$$

where d is the distance from 1.84 meters below the point of maximum star density to the aquifer (11.3 meters).

Due to the lack of cylindrical symmetry of the modeled CASIM geometry, it was necessary to define separate regions of soil beneath the target/dump and to determine the star density in those regions by dividing the total stars produced in the regions by the volume of the regions. Two rectangular regions were defined below the beam line. Each of them was 100 cm (l) x 30 cm (w) x 15 cm (h). The first was centered directly below the 100 cm long tungsten block immediately beneath the concrete floor slab. This was beneath the thinnest part of the steel shielding surrounding the target and is in the region where the maximum star density in the unprotected soil would be expected. The second region was beneath the upstream end of SELMA, immediately downstream of the first region, and also centered below the beamline immediately beneath the floor slab. However, the floor slab is about three feet lower in elevation at this location and there is also an additional 45 cm of concrete on which the SELMA magnet rests, so this region would be expected to have a reduced star density. These regions are indicated in figure 7b.

The total number of stars per incident proton produced in the region beneath the dump was 1.9 $\times 10^{-3}$ ($\pm 0.6\times 10^{-3}$) stars, based on a CASIM run with 100,000 incident particles at 800 GeV. Dividing by the region's volume of 4.5×10^{4} cm^3 gives a maximum star density of 4.2×10^{-8} stars per cm^3 per proton . Solving for the initial and final concentrations for H^3 and Na^{22} in terms of N_p using equations 3 through 6, substituting the results into eq. 2 and solving for N_p gives a maximum of 4.4×10^{18} ($\pm 1.3\times 10^{18}$) protons per year, averaged over three years.

Conclusion

The beam dump shielding design as modeled for the E872 experiment in PW8 is adequate for the number of protons on target required by the experiment using either the Single Resident Well Model or the Concentration Model.

APPENDIX 1

CASIM FORTRAN Geometry File

```
      SUBROUTINE CASIMGEOM
C
C       USER SUBROUTINE DESCRIBING PROBLEM GEOMETRY
C
C       GIVEN ( X,Y,Z ) IN CM, ENTRY USRGEOM RETURNS
C               MATERIAL INDEX N (GE.-1  AND LE.9)
C               MAGNETIC REGION INDEX M (GE.0)
C
C       CONVENTIONS NM=-1 VACUUM WITH MAGNETIC FIELD PRESENT
C                       0 VACUUM  NO  MAGNETIC FIELD PRESENT
C                     1-9 MATERIALS
C                      99 Outside of defined geometry
C
C       MAGNETIC FIELD INDEX M MUST CORRESPOND TO INDEX IN
C       SUBROUTINE USRFIELD WHERE B VECTOR IS DEFINED AS A FUNCTION OF
C       LOCATION
C
      IMPLICIT NONE
      SAVE
C
C========================= Include HIBI.CIN ==========================
C HIBI
C
      REAL ZLIM, RLIM
C
      COMMON/HIBI/ ZLIM, RLIM
C=========================== End Include =============================
C
      INTEGER N, M, KT
C
      REAL X, Y, Z, XM, YM, ZM, R
      REAL AX, AY
C
C-------------------------------------------------------------------
C Hard-wire the limits to your geometry here.  This section gets
C called once per Job.  The 'SAVE' above insures that values set
C are retained after the subroutine is exited.
C
      ZLIM=3400.
      RLIM=750.
C
      RETURN
C
C-------------------------------------------------------------
C
C This section gets called whenever CASIM wants to check what material
C a particle is in, given X, Y, Z
C
      ENTRY USRGEOM(X,Y,Z,N,M,KT,XM,YM,ZM)
C
C    The E-872 beam dump model geometry for a Casim calculation of ground water
C    activation.
C
C    Included are a full and half density tungsten target surrounded by steel
C    shielding followed by the SELMA magnet followed by MuSweep 2
C    followed by three long pieces of steel shielding/absorber for the
C    detector. The beam is targeted on the full density side of the target.
C
C    Steel shielding out to a distance from the beam line of 120 cm has been
C    added around the target/dump on the top and sides.  Steel out to
C    a distance of 91.44 cm from the beam elevation has been added below the
C    target/dump. This is the maximum allowed due to the floor elevation.
C    The target is recessed by 100 cm along the beam direction
```

```
C    into the steel that surrounds it.
C
C    10 cm thick layer of tungsten (20 cm wide) is installed  below the dump
C    to compensate for the reduced steel thickness. (COMMENTED OUT FOR THIS RUN)
C
C    The Selma magnet is modeled as a steel block 220 cm x 290 cm x 700 cm
C    with a filled gap. The vertical width has been increased from 190 to 220
C    to see effect on ground water by adding a foot of steel on top.
C    45 cm of concrete has been added on the floor below SELMA.  This is a base
C    block on which SELMA rests.
C
C    Mu Sweep 2 is modeled as a picture frame magnet with a filled gap.
C
C    Three pieces of passive steel shielding downstream of the dump are included
C    The first is  800 cm x 280 cm x 80  cm
C    The second is 400 cm x 280 cm x 140 cm
C    The third is  300 cm x 280 cm x 280 cm
C
C    The PW8 Experiment hall modeled as a concrete enclosure with
C    18 inch thick concrete ceiling walls and floor.
C
C
C    N=1         Steel
C    N=2         Copper  (not used in currently specifed geometry)
C    N=3         Tungsten
C    N=4         Soil
C    N=5         Concrete
C
C                 X,Y,Z          ARE "LAB" COORDINATES
C                 XM,YM,ZM       ARE "MAGNET" COORDINATES
C                 IN PRESENT EXAMPLE THEY ARE IDENTICAL
C
C Keep these 5 lines here, always.  CASIM needs the counter KT, and needs to
C know when the counter has gone beyond it's limit
      KT=KT+1
      IF(KT.GT.10000) THEN
         N = 99
         RETURN
      ENDIF
C
C Set N=0 until we know if we are in some other material
      N=0
C
C Set M=0 until we know if we are in a field region
      M=0
C
C First check the limits of the geometry and exit if outside them
      R = SQRT(X*X + Y*Y)
      IF (R .GT. RLIM) THEN
         N = 99
         RETURN
      ENDIF
C
      IF (Z.LE.0. .OR. Z.GT.ZLIM) THEN
         N = 99
         RETURN
      ENDIF
C
C
C Set "Magnet" coordinates the same as Lab coordinates
      XM=X
      YM=Y
      ZM=Z
C
      AX=ABS(X)
      AY=ABS(Y)
C
C     Define Geometry
C
```

```
C     Assume vacuum everywhere
      N=0
C
C     Define steel shielding/hole/tgt(full and half density)
      IF(Z.LT.200.) THEN
        IF(AY.LT.120..AND.(X.LT.120..AND.X.GE.-91.44)) N=1  !Steel
        IF(AY.LT.10..AND.(X.LT.20..AND.X.GE.-6 .)) N=0      !Hole
        IF(Z.GE.100.) THEN
          IF(AX.LT.5..AND.AY.LT.2.5) N=3 !Full density tungsten tgt
          IF(AX.LT.5..AND.(Y.GE.2.5.AND.Y.LT.7.5)) THEN
C           Half-density tungsten tgt
            IF(Z.LT.110.) N=3
            IF(Z.GE.120..AND.Z.LT.130.) N=3
            IF(Z.GE.140..AND.Z.LT.150.) N=3
            IF(Z.GE.160..AND.Z.LT.170.) N=3
            IF(Z.GE.180..AND.Z.LT.190.) N=3
          ENDIF
C*****    IF(AY.LT.10..AND.(X.LT.-6..AND.X.GE.-16.)) N=3 !Tungsten below dump
        ENDIF
      ENDIF
C
C     Define Selma magnet as solid steel block
      IF(Z.GE.200..AND.Z.LT.900.) THEN
C*****  IF(AX.LT.125..AND.AY.LT.145.) N=1 !30 cm steel added on top and bottom
C*****  IF(AX.LT.95..AND.AY.LT.145.) N=1  !nominal SELMA dimensions
        IF((X.LT.125..AND.X.GE.-95.).AND.AY.LT.145.) N=1 !30 cm steel on top only
        IF((X.LT.-137.9.AND.X.GE.-189.9).AND.AY.LT.145.) N=5 !Extra concrete below SELMA
      ENDIF
C
C     Define MuSweep 2
      IF(Z.GE.1300..AND.Z.LT.1800.) THEN
C       Top/Bottom Steel
        IF(AX.LT.120..AND.AX.GE.30.) THEN
          IF(Y.LT.35..AND.Y.GE.-325.) N=1
        ENDIF
C       East/West Steel
        IF(AX.LT.30.) THEN
          IF(AY.LT.35.) N=1  !East
          IF(Y.LT.-171..AND.Y.GE.-325.) N=1 !West
        ENDIF
      ENDIF
C
C     Passive Steel 1
      IF(Z.GE.1900..AND.Z.LT.2700.) THEN
        IF(AX.LT.140..AND.AY.LT.40.) N=1
      ENDIF
C
C     Passive Steel 2
      IF(Z.GE.2700..AND.Z.LT.3100.) THEN
        IF(AX.LT.140..AND.AY.LT.70.) N=1
      ENDIF
C
C     Passive Steel 3
      IF(Z.GE.3100..AND.Z.LT.3400.) THEN
        IF(AX.LT.140..AND.AY.LT.140.) N=1
      ENDIF
C
C
C     Roof Slab
      IF(X.LT.513.1.AND.X.GE.467.4) THEN
        IF(AY.LT.457.2) N=5
      ENDIF
C
C     Side Walls
      IF(AY.GE.457.2.AND.AY.LT.502.9) THEN
        IF(Z.LT.154.3) THEN
          IF(X.LT.513.1.AND.X.GE.-137.2) N=5 !Upstream Side Walls
        ELSE
          IF(X.LT.513.1.AND.X.GE.-228.6) N=5 !Downstream Side Walls
```

```
        ENDIF
      ENDIF
C
C     Floor Slabs
      IF(AY.LT.457.2) THEN
        IF(Z.LT.154.3) THEN
          IF(X.LT.-91.44.AND.X.GE.-137.2) N=5 !Upstream Slab
        ELSEIF (Z.GE.154.3.AND.Z.LT.200.) THEN
          IF(X.LT.-91.44.AND.X.GE.-228.6) N=5 !Transition Slab
        ELSE
          IF(X.LT.-182.9.AND.X.GE.-228.6) N=5 !Downstream Slab
        ENDIF
      ENDIF
C
C     Soil Regions outside enclosure walls out to RLIM
      IF(AY.GE.502.9) N=4                !Sides
      IF(AY.LT.502.9.AND.X.GE.513.1) N=4 !Top
      IF(Z.LT.154.3) THEN
        IF(X.LT.-137.2) N=4              !Upstream below floor
      ELSE
        IF(X.LT.-228.6) N=4              !Downstream below floor
      ENDIF
C
C     Special soil region below tungsten dump for concen. model est.
      IF(Z.GE.100..AND.Z.LT.200.) THEN
        IF(X.LT.-137.2.AND.X.GE.-152.2) THEN
          IF(AY.LT.15.)  N=6  !Special soil bin below dump
        ENDIF
      ENDIF
C
C     Special soil region below Selma for concen. model est.
      IF(Z.GE.200..AND.Z.LT.300.) THEN
        IF(X.LT.-228.6.AND.X.GE.-243.6) THEN
          IF(AY.LT.15.)  N=7  !Special soil bin below Selma
        ENDIF
      ENDIF
C
      RETURN
      END
```

Appendix 2

CASIM Input Data File

```
! The first lines hold the list of customized files you need to run
! your CASIM geometry.  The two listed here are the MINIMUM you need, and
! should always be there.  If you gave the CASIMGEOM file a customized name,
! then you need to edit that name here in place of CASIMGEOM.FOR.
! Add more files as needed, add them as new lines below these two,
! 1 file-name per line, left-justified.
CASIM.FOR
E872DUMPGEOM.FOR
E872DUMPGEOMVIEW.FOR
E872CASPRIMAR.FOR
!!!!!!!!!!! The CASIM Input Deck information starts below this line !!!!!
        1      800.      0.3         0.15      0.15       1.0       100000
   0    0    0    2 635657139  1.E-18     10.0       15.0
        7
   26.0      56.0       7.87      287.       1.80       4.97        0.386
   29.0      63.5       8.96      314.0      1.43       5.60        0.510
   74.0     183.9      19.30      746.0      0.35       7.39        1.139
   10.7      21.7       2.25      135.8     12.25       3.62        0.163
   11.4      22.9       2.40      142.7     10.89       3.69        0.171
   10.7      21.7       2.25      135.8     12.25       3.62        0.163
   10.7      21.7       2.25      135.8     12.25       3.62        0.163
!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
!!!!! the information below is ignored by all CASIM command proceedures
!!!!! but is provided to list the standard parameters for materials used
!!!!! by most CASIM geometries.  Copy lines below into the Input Deck above
!!!!! as needed.  Do NOT copy the preceeding material ID line.  The order
!!!!! in which they are listed does not imply any assumptions within CASIM.
! Fe  Iron
   26.0      56.0       7.87      273.0      1.76       5.39        0.510
! Cu  Copper
   29.0      63.5       8.96      314.0      1.43       5.60        0.510
! Cu/H2O combined, for water cooled magnet coils
   29.0      63.5       8.30      314.0      1.54       5.60        0.510
! Brass
   29.0      63.5       9.00      314.0      1.56       5.60        0.510
! Al  Aluminum
   13.0      27.0       2.70      166.0      8.90       4.20        0.215
! Be  Berylium
    4.0       9.0       1.85       60.0     35.30       3.47        0.051
! W   Tungstun
   74.0     183.9      19.30      962.0      0.35       7.21        1.400
! Pb  Lead
   82.0     207.2      11.35     1070.0      0.50       7.50        1.540
! Soil
   11.0      23.0       2.00      139.0      8.90       3.45        0.170
! Concrete
   11.0      23.0       2.40      139.0     10.70       4.00        0.220
```

INPUT DATA TABLE
Stars per incident particle 1.03E-01
Protons per year 2.00E+18
Distance to acquifer (feet) 37

ISOTOPE DATA TABLE	H3	Na22
Mean lifetimes (years)	17.7	3.74
Migration rate (ft per year)	7.2	3.2
Allowed concentrations (pCi per ml)	20	0.2
Leachable atoms per star	0.0675	0.0027

OUTPUT DATA TABLE	H3	Na22
Concentrations (pCi per ml)	9.09	0.10
Ratio to allowed concentration	0.45 OK	0.52 OK

Combined Ratio to EPA 4 mrem limit	0.98 OK

All OK?	YES

Table 1 - Single Resident Well Model spreadsheet results



Figure 1a - Cross section through target/dump shielding region (dimensions in cm)



Figure 1b - Elevation view through target/dump shielding region (dimensions in cm)



Figure 2 - Tungsten target geometry (dimensions in cm)



Figure 3 - SELMA magnet geometry (dimensions in cm)



Figure 4 - MuSweep 2 cross section (dimensions in cm)



Figure 5 - Muon absorber steel (dimensions in cm)



Figure 6a - PW8 Cross Section - upstream (dimensions in cm)



Figure 6b - PW8 cross section - downstream (dimensions in cm)



Figure 7a - Plan view of full geometry



Figure 7b- Elevation view of full geometry